

パーソナル・コンピュータのベーシック(1)

小林 竜一

はじめに

この連載の目的は、最近パーソナル・コンピュータ(マイコン)が20~100万円で購入することができ、個人用に専有して使え、しかもひと昔まえの大型コンピュータの能力をもっているため、かなりの普及がなされているので、そのマイコンで使われている言語 BASIC(Beginner's All-purpose Symbolic Instruction Code)について解説をすることである。BASIC は現在はポピュラーな言語だと思われているが、意外にまだ使ってよい方が使っておられないこともあるので紙面を借りその利便性を示すことにしたい。また、BASIC はマイコンの種類によっていろいろの変形文法が見られるので本稿では、主として価格100万円以下のマイコンのBASICの比較を行なうことにしたい。

なお、BASIC は文献[1]の表紙カバーの背表紙に実に明快な紹介文が載っているのでそれを以下に引用させていただきます。

ベーシックとは

1965年アメリカのダートマス大学で開発された(BASIC)言語とその処理系は、初心者にとってたいへんつつきやすく、熟練者にとって使いやすい優秀なシステムだと定評がある。その特色は、①FORTRANに似ている、②ALGOLのよいところがとりいれてある、③入力データのケタを合わせる必要がない(自由書式)、④出力の書式を特に指定しなくても印字される(標準書式)、⑤かなり自由に「見出し」を入れたり間隔をつめたりもできる、⑥整数と実数の区別にあまり気を使わないでよい、⑦リスト(一元表)とテーブル(二元表)が使える、⑧行列演算が備わっていて、それが強力かつ柔軟である、⑨プログラムを修正して再び実行することがきわめて容

易にできる、⑩人間と計算機との対話に適する機能がある。

BASIC ははじめ大型コンピュータをTSS 端末(電話回線を利用してデータを送受する装置)を通じて使用するために考えられた。ところがこの言語が計算機と人間との対話型のプログラミングに適していることから、最近のマイコンの普及にしたがって、一部の文法をマイコン用に追加修正して使われることになって、マイコンといえばベーシックという状態となったのである。おそらく現在ではTSS 端末よりマイコンの上で走っているBASICのほうが圧倒的に多いのではないだろうか。

本連載はまず基本的なBASICとしてその創作者のKEMMENY & KURTZ [1]のものを要点だけ平易に説明する。その後以下に示す各社のマイコンのBASICの特長を述べることにする。

- IEE 社(MITS 社日本代理店) ALTAIR BASIC
- COMMODORE JAPAN 社 コモドル BASIC
- シャープ社 BASIC mz-80 シリーズ
- TEXAS INSTRUMENTS 社 TI-99/4 ベーシック
- 横河ヒューレットパッカード社 HP-85 ベーシック
- TANDY 社(タンディラジオシャック) TRS-80 I ベーシック(レベルII)

以上6社以外にも資料が手に入れば追加解説を行なう。なお、これはすべて各社のユーザース・マニュアルなどを入手して行なうもので、実際にマイコンを使用して確認していないので筆者の不注意な記述が皆無ではないおそれがある。この点をあらかじめお許しを願ひ、資料を提供して下さった各社にお礼を申し上げる。読者がお気づきの誤りについては筆者にお知らせ給われれば後の誌面をお借りして訂正させていただきたいと思う。

1. ベーシック(BASIC)

まず最初に基本的な(初めて世に出たときの)BASICについて解説する。実例についてBASICでプログラム

こばやし りゅういち 立教大学

```

10 REM MEAN AND SUM OF SQUARES
20 DATA 3.1,2.9,2.7,3.0,3.1
30 DATA 4.0,3.6
40 LET M=0
50 FOR I=1 TO 7
60 READ X
70 LET M=M+X
80 NEXT I
90 LET M=M/7
100 PRINT "MEAN ",M
110 LET S=0
120 RESTORE
130 FOR I=1 TO 7
135 READ X
140 LET S=S+(X-M)2+(X-M)
150 NEXT I
160 PRINT "SUM OF SQUARES ",S
170 STOP
180 END

```

図 1 平均値と平方和のプログラム

してみよう。

[実例 1] 次に示すデータの平均値を出せ。またそのあとで平方和を出せ。

データ：3.1, 2.9, 2.7, 3.0, 3.1, 4.0, 3.6

図1のプログラムを解説する。まずREM文である。これは注釈(Remark)であって何も実行はされない。プログラムの中に何をしておくか書いておく憶え書きにつかう。次にDATA文である。これはデータをしまう領域を示す。図1のように書いても、または、

```
20 DATA 3.1, 2.9, 2.7
```

```
30 DATA 3.0, 3.1, 4.0, 3.6
```

と書いても同じである。次にLET文である。次の式の右辺の値を左辺に代入することを示す。この場合Mという名前の記憶場所に0を代入する(メモリークリア)。次はFOR文である。これは計算や入出力などの実行文(それはこのFOR文の下に書いてある。)を繰返し実行することを宣言する。ただし、I=1から7まで、つまり、

```
I=1, 2, 3, 4, 5, 6, 7
```

の場合について繰返し実行する。次にREAD文がある。これはまず最初(I=1のとき)はXという名の記憶場所にDATA領域の最初の数3.1を読み込む。次のLET文でMの中にある数とXの中にある数を加えてこれを新しくMの中に入れる(3.1がMの中に加算)。次にNEXT文がある。これでI=2としてFOR文の次の文つまりREAD文に帰っていく。するとこのREAD文がデータ領域の2番目の数を読んでXの中に入れる。ここで2番目というのはIが2だからではなく、READ文は使われる毎にデータ領域を先頭から次々に読み出してくるからである。1度データ領域の数が1つ読まれると、次にREAD文はその次の数を読む体勢となるのである。

MEAN	3.2	
SUM OF SQUARES		1.2

図 2 図1のプログラムの出力例

かくして80番の行番号までを7回繰返すと90番の行にくる。このLET文によってMの中にデータの平均ができる(7でわるから)。次のPRINT文でMEAN、(はブランクを示す)と印刷しそのあとに平均値を印刷する。書式は処理系(インタープリタ、このベーシックのプログラムの意味を調べつつ実行するプログラム)が適当に決めてくれる。次に110番の行番号でSという名前の記憶場所に0を入れる。その次の120番のRESTORE文で、これはデータ領域のポインタを最初にもどす。ここでポインタとは次のREAD文がデータ領域の中のどの数を読むべきかを指し示すものこと、この場合RESTORE文でポインタは3.1という数を指し示す。そこで140番の行番号でのXははじめは3.1である。次にFOR NEXTのループが図るときはXは2.9である(以下同様)。かくしてループを7回まわって次のPRINT文にいき、SUM OF、SQUARES、と印刷してSの内容を適当な書式で印刷してくれる。そして次のSTOP文で計算は終りとなる。次のEND文はプログラムの終りを示す。

このプログラムを実行させる(RUNというコマンド(計算機に与える命令)を使うのが普通である)。図2のような出力が得られる。

◆行番号は普通10番とびにつける。もし2つの行の間に新しい行を入れたかったら、いつでも(それらの行を書き終ったあとでも(実行のまえなら)その中間の番号の行を追加して、その行に入れることができる。

◆不必要な行(消したい行)は行番号だけ入れると消える(行番号のあとに復帰改行のキーを押す)。

◆LET文のLETを省略してもよいBASICが最近は多くなってきている。

◆STOP文は計算の終りを示し、END文はベーシックのプログラムの終りを示す。

◆140番の行はS=S+(X-M)²と書いてもよい。ただし機種により↑はハであったり**であることもある。

◆DATA文はプログラムのどこにあってもよい。最後のほうでも構わない(END文よりは前とする)。

◆上例ではPRINT文の中で印刷すべきものをコンマ、で区切っているが、セミコロン;で区別されて印刷される文字や数字が少しまつて(狭いところに)印刷できる。

◆変数名は英字1文字がまたはその外に数字1文字をつ

```

10 REM MEAN AND SUM OF SQUARES
20 DIM X(100)
30 INPUT N
40 IF N<=0 THEN 110
50 FOR I=1 TO N
60 INPUT X(I)
70 NEXT I
80 GOSUB 200
90 GOSUB 300
100 STOP
110 PRINT "ERROR"
120 GO TO 30
200 M=0
210 FOR I=1 TO N
220 M=M+X(I)
230 NEXT I
240 M=M/N
250 PRINT "MEAN ";M
260 RETURN
300 S=0
310 IF I=1 THEN 360
320 FOR I=1 TO N
330 S=S+(X(I)-M)*(X(I)-M)
340 NEXT I
350 PRINT "SUM OF SQUARES ";S
360 RETURN
99999 END

```

図 3

けたものだけが許される。

上述の例では計算のためのデータがプログラムの中に書かれていたが、これをキーボードから入力するようにし、また平均値を計算するサブルーチン（サブルーチンとは図3の中の200番の行から260番の行までのようなものをいう）、平方和を計算するサブルーチンを別にかくと次の図3のようなプログラムとなる。なおデータの数も可変としNという名の記憶場所に入れるものとする。

◆INPUT文はキーボードから数字(または文字)を読み込むときに使用する。

◆文字を読み込む記憶場所の名前には最後に\$をつける。たとえばA\$, B\$, C3\$など(拡張されたベシック)。

```

30 INPUT A$
40 PRINT A$

```

などと入出力文が書ける。もちろん以下のようにも書ける。

```

100 READ A$, B$, C$
110 PRINT A$; B$; C$
:
300 DATA KOBA, YA, SHI

```

文字列の長さは60字以内でないといけない。

◆文字列データ領域のポインタを元にもどすには RES-TORE\$文を使う。

```

??
??3.1
??2.9
??2.7
??3.0
??3.1
??4.0
??3.6
MEAN 3.2
SUM OF SQUARES 1.2

```

図 4 出力例

◆GOSUB文 サブルーチンをよぶ命令である。

◆RETURN文 サブルーチンに必ず1つ以上ないといけない。よばれた所の次の文へ帰る。

◆IF THEN文 IFとTHENの間に書かれている条件が満たされているときTHEN文のうしろに書かれた行番号のところへ飛んでいく。

IF(条件式) THEN(行番号)

◆RESEQUENCEというコマンドが使える。これは行番号をつけ直す命令で、これによって行についた番号と、それに関係のあるIF THEN文とGO TO文の番号をつけ直す。RESEQUENCEとだけ入れると10番を最初の行にし、あとは10番ずつ加えた行番号をつけてくれる(そういう機械が多い)。

◆プログラム全文を印刷させるのにLISTというコマンドが使える(そういう機械が多い)。

◆DEF文によって新しい関数を定義できる。たとえば、

```
20 DEF FNA(X,Y)=SQR(X↑2+Y↑2)
```

とするとSQRが平方根を表わす関数なので、これで

$$f(X, Y) = \sqrt{X^2 + Y^2}$$

という関数が定義される。関数の名前は英字3字で最初の2字はFNと決められている。

◆BASICでは以下に述べる11の関数が使える(もちろんDEF文なしで使ってよい)。表1を見よ。

◆関係演算子(IF THEN文で使う)には以下の6種類がある(表2)。

2. BASICの行列演算

ベシックでは行列演算が簡単にできるのが1つの特長となっている。これは次の例題を見ればすぐ理解できるであろう。

[実例2] 次の行列A, BについてC=A+B, D=AB'(ダッシュは転置)を計算し、さらにD⁻¹を求めよ。

$$A = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} \quad B = \begin{bmatrix} 7 & 8 & 9 \\ 0 & 1 & 2 \end{bmatrix}$$

上述のプログラム例と出力を見ていただければ行列演算

表 1 標準関数

関数記号	数学記号	備 考
SQR()	$\sqrt{\quad}$	SQR(2)だと1.4141...
SIN()	sin	正弦関数
COS()	cos	余弦関数
TAN()	tan	正接関数
ATN()	\tan^{-1}	逆正接関数
LOG()	\log_e	自然対数
EXP()	$e(\exp)$	自然対数の底 e のべき乗
ABS()		絶対値
SGN()	(読み方: シグナム)	符号 (注1)
INT()		小数点以下切捨て
RND()		一様乱数 (注2)

注1) SGN(x)は $x > 0$ のとき +1, $x = 0$ のとき 0, $x < 0$ のとき -1 となる。

注2) RND(x)は 0~1.0の区間上の一様乱数を発生する関数である。しかし x (これを引数(ヒキヌウ)という)の値のとりあつかいについては標準的な約束では、 x は何でもかまわないこととしてあるが、処理系によっては x が 0 のときや正や負のときに別の働きをもたらすものもあるので使用するコンピュータのマニュアルをよく調べて使用することが必要である。

が容易に行なえることが直ちにわかるであろう。

◆上の例に出てこない演算を以下に例でもって示す。

- 減算 310 MAT C=A-B
- スカラー一倍 350 MAT C=(2.0)A (2倍の例)
- ゼロ行列 370 MAT C=ZER (Cの要素が全部0となる)
- 1化 390 MAT C=CON(3,4) (Cの要素をすべて1とする)

(注) Cの大きさは 4×5 となる (ベーシックでは添字 (括弧の中の数) は 0 から動く。つまり、たとえば DIM X(3)は X(0), X(1), X(2), X(3) という3つの場所を宣言することになる。

単位行列 400 MAT C=IDN(N,N) (N次の単位行列)

◆DELETE, EXTRACT, MERGE, WEAVE というようなコマンドが使用できる機械がある (参考文献[2], 120頁~124頁)

◆プログラム作成中間違ったら、同じ番号をつけた行をはじめから入れ直せばよい。同じ番号の行が何行かあると最後に入った行が生きる。

◆消去したい行が出たらその行についている番号だけを入れ (復帰改行をする) ればよい。

```

10 REM MATRIX CALCULATION
20 DIM A(2,3), B(2,3), C(2,3), B1(3,2), D(2,2), E(2,2)
30 DATA 1,2,3,4,5,6,7,8,9,0,1,2
40 MAT READ A,B
50 MAT PRINT A
60 MAT PRINT B
70 MAT C=A+B
80 MAT PRINT C
90 MAT B1=TRN(C)
100 MAT D=A*B1
110 MAT PRINT D
120 MAT E=INV(D)
130 MAT PRINT E
140 STOP
99999 END
    
```

図 5

表 2 関係演算子

関係演算子	数学記号	備 考
<	<	より小
<=	\leq	以下
=	=	等しい
>	>	より大
>=	\geq	以上
<>	\neq	不 等

3. ベーシックの勉強法

BASIC を初めて勉強しようと思われた方は、ためらわず、BASIC の使用できる機械を買われるとよい。まず、一番安価な機械はシャープの PC-1210 または PC-1211で、これなら3, 4万円で買うことができ、すぐに使いこなすことができる。もう少し予算のある方はマイクログンピュータ (各社から出ている) を20万円位から買うことができる。この場合は機種によってそれぞれ特徴があり、また1つの機種に決めたらほかの機種に代えることはなかなか難しいから慎重に決めなければならない。これらは各社がショールームを各地に設けているのでそこで実際に機械に手をふれてみるとよい。20~100万円で買えるパーソナル・コンピュータはたとえば表3のようである。

◆S-100バスのマイコンはS-100バス規格の各種のボード(入出力ボード, メモリーボード, 発音回路のボード, フロッピーディスクコントローラなど)を必要に応じて買って挿入することができる。

◆OS(オペレーティングシステム)に CP/M が使っているものは、この OS のもとで動く多くのプログラム(FO-RTRAN コンパイラ, BASIC コンパイラ, COBOL コンパイラ etc) を買って使うことができて便利である。

◆独自のバスライン, 独自の OS の場合には、その機械

の発売元からサポートされているボードやソフトしか利用できないので、その点が不便である。もちろん自分でプログラムを作るなら問題はないが、ソフトを作るのは非常に時間がかかることを考えておくべきである。

参 考 文 献

- [1] KEMMENY & KURTZ ベーシック入門 森口繁一監訳 尾崎義雄・神山武共訳 共立出版社刊
- [2] FARINA タイムシェアリング・プログラミング=BASIC による 関根智明訳 培風館刊(絶版)
- [3] SNELL 確率と BASIC 池浦外2氏訳 共立出版社刊
- [4] 伊藤, 木下 タイムシェアリング用言語 竹内書店
- [5] 前田英明 マイコンのための BASIC の使い方 共立出版社刊
- [6] 雑誌 トランジスタ科学 CQ出版社発行

1	2	3
4	5	6
7	8	9
0	1	2
8	10	12
4	6	8
50	8	
122	17	
-.1349206	.0634921	
.968254	-.3968254	

図 6 出力例

表 3 パーソナルコンピュータ

発 売 元	商 品 名	ハード規格	ソフト(OS)
IEE 社	ALTAIR	S-100バス	CP/M
コモドールジャパン	PET	独自	独自
シャープ	PC-3100	独自	独自
日本電気	MZ-80 シリーズ	独自	独自
ソード	PC-8001	独自	CPM
タンディ	M200 シリーズ	独自/S-100バス	独自
YHP	TRS-80	独自	独自
日立	HP-85	独自	独自
沖	ベーシックマスター	独自	独自
TEAC	IF800	独自	独自
テキサス・インスツルメンツ・ジャパン	PS-80	独自	独自
APPLE COMPUTER	TI99/40	独自	独自
HORIZON COMPUTER	APPLE II	独自	独自
日本クロメンコ社	HORIZON*	S-100バス	CP/M
インターナショナルサイエンティフィック	Cromemco*	S-100バス	独自(CP/Mコンパチブル)
常陽	KAISER Z80	S-100バス	CP/M
アドテック	SPIRITS	独自	CP/M(可)
日本パーソナルコンピュータ社	ORANGE	独自	独自
アスターインターナショナル	DPS-1	S-100バス	CP/M
Cannon	COSMOS/II*	S-100バス	CP/M
精工舎	BX-30	独自	独自
国際データ機器株式会社	SEIKO 8500*	独自	独自
	PDS-V*	不明	CP/M

(注) *印がついたものは100万を越すもの。